Weighted nonlinear fit

Created using Maple 14.01
Jake Bobowski

```
> restart;
  with(StringTools) :
  with(Statistics) :
  with(plots) :


  FormatTime("%m-%d-%Y, %H:%M");
```

"03-25-2015, 16:17" **(1)**

In this example, *V*r/*V*o data from an *LRC* series circuit will be plotted as a function of angular frequency ω.

This Maple input enters a list of the calculated ratio of *V*r/*V*o from measurements made in the lab. Also enter the uncertainty in the voltage ratio. Both *V*r and *V*o have independent uncertainties. Have to use propagation of errors to find Δ*Vratio*.

```
> Vratio := [.0198, .067, .117, .185, .331, .450, .573, .689, .718, .714, .704, .670, .631, .549, .412,
      .318, .249, .186, .138];
  σVratio := [.001, .002, .003, .003, .005, .006, .007, .008, .008, .008, .008, .008, .008, .007,
      .006, .004, .005, .003, .003];
```

Vratio := [0.0198, 0.067, 0.117, 0.185, 0.331, 0.450, 0.573, 0.689, 0.718, 0.714, 0.704, 0.670, 0.631, 0.549, 0.412, 0.318, 0.249, 0.186, 0.138]

σVratio := [0.001, 0.002, 0.003, 0.003, 0.005, 0.006, 0.007, 0.008, 0.008, 0.008, 0.008, 0.008, 0.008, 0.007, 0.006, 0.004, 0.005, 0.003, 0.003] **(2)**

Calculate the weights for the fit [(1/Δ*Vratio*)^2].

$$> Yweights := \left[seq\left(\frac{1}{\sigma Vratio[i]^2}, i = 1..nops(\sigma Vratio) \right) \right];$$

$Yweights := [1.\ 10^6, 2.500000000\ 10^5, 1.111111111\ 10^5, 1.111111111\ 10^5, 40000.00000, 27777.77778, 20408.16327, 15625.00000, 15625.00000, 15625.00000, 15625.00000, 15625.00000, 15625.00000, 20408.16327, 27777.77778, 62500.00000, 40000.00000, 1.111111111\ 10^5, 1.111111111\ 10^5]$ **(3)**

Enter the frequency of the generator in Hz. I'm assuming that the fractional error in the frequency is much less than that of the voltage ratio. Calculate the angular frequency ω.

```
> f := [3000, 10000, 15000, 20000, 25000, 28000, 30000, 32000, 33000, 34000, 35000, 36000,
      37000, 39000, 43000, 47000, 52000, 60000, 70000];

  ωdata := evalf(2·Pi·f);
```

f := [3000, 10000, 15000, 20000, 25000, 28000, 30000, 32000, 33000, 34000, 35000, 36000, 37000, 39000, 43000, 47000, 52000, 60000, 70000]

(4)

$\omega data := [18849.55592, 62831.85308, 94247.77962, 1.256637062\ 10^5, 1.570796327\ 10^5, 1.759291886\ 10^5, 1.884955592\ 10^5, 2.010619298\ 10^5, 2.073451152\ 10^5, 2.136283004\ 10^5, 2.199114858\ 10^5, 2.261946710\ 10^5, 2.324778564\ 10^5, 2.450442270\ 10^5, 2.701769682\ 10^5, 2.953097094\ 10^5, 3.267256360\ 10^5, 3.769911184\ 10^5, 4.398229716\ 10^5]$ **(4)**

Now plot *Vratio* vs ω.

```
> DataPlot := ScatterPlot( ωdata, Vratio, yerrors = σVratio, axes = boxed, view = [0 .. 5e5, 0 .. 0.8], labels = [typeset("ω (rad/s)"), typeset("voltage ratio")], labeldirections = ["horizontal", "vertical"], tickmarks = [7, 8], axesfont = [Times, 12], labelfont = [Times, 14], axis = [gridlines = [thickness = 1]], symbolsize = 10, symbol = solidcircle, thickness = 2)
:
display(DataPlot);
```



Now, fit the data to a nonlinear function. We will use the Maple function *NonlinearFit.*

The format of the command below is *LinearFit*(fcn, x data, y data, variable, acceptable ranges of parameters, weights, output options). With the options selected, the output gives the fit function and the best fit parameters (*A*, *g*, and ω*o*).

> $fcn := NonlinearFit\left(\frac{A}{\left(1+\frac{\omega^2}{g^2}\left(1-\frac{\omega o^2}{\omega^2}\right)^2\right)^{\frac{1}{2}}}, \omega data, Vratio, \omega, parameterranges = [A = 0 ..2, g = 0 ..300000, \omega o = 0 ..500000], weights = Yweights, output = [leastsquaresfunction, parametervalues]\right);$

$$fcn := \left[\frac{0.729649817392542}{\sqrt{1 + 2.33105403776590\ 10^{-10}\,\omega^2\left(1-\frac{4.62834279414143\ 10^{10}}{\omega^2}\right)^2}}, [A = 0.729649817392542, g = 65497.3651414017, \omega o = 2.15135836023231\ 10^{5}]\right] \quad (5)$$

Now plot the fit function defined by *fcn* over the entire frequency range. Give the plot the name *FitPlot.*

> $FitPlot := plot(fcn[1], \omega = 0 .. 450000, colour = green) :$
> $display(FitPlot);$



Use *display* to show both *DataPlot* and *FitPlot*.

> *display*(*DataPlot*, *FitPlot*);



Note that Maple does not report the uncertainties in the bet-fit parameters. We have to do some work to estimate these uncertainties. The first step is to calculate the Chi-Squared value using the best-fit parameters which should correspond to the minimum Chi-Squared value (ChiSqZero). The 'ω' notation clears any value previously assigned to ω.

> $ChiSq := 0$:
for i **from** 1 **to** $nops(Vratio)$ **do**:
$\omega := \omega data[i]$:

$$ChiSq := ChiSq + \left(\frac{(Vratio[i] - fcn[1])}{\sigma Vratio[i]}\right)^2 :$$

end do:
$ChiSqZero := ChiSq$;
ω :='ω':

$$ChiSqZero := 73.2231847528438 \qquad \textbf{(6)}$$

The next step is to vary one of the parameters away from the value that minimizes ChiSq to see how ChiSq depends on the parameter value. We expect this variation to result in a quadratic dependence. As an example, we'll try finding the uncertainty in the resonance frequency ωo. Note, however, the same procedure can be applied to each of the parameters (A, g, ωo).

First, load the best-fit parameters determined from the fit into memory.

```
> Astar := rhs(fcn[2, 1]);
  gstar := rhs(fcn[2, 2]);
  ωostar := rhs(fcn[2, 3]);
```

$$Astar := 0.729649817392542$$
$$gstar := 65497.3651414017$$
$$\omega ostar := 2.15135836023231\ 10^{5} \quad \textbf{(7)}$$

Next, pick a suitable step size by which to vary ωo. This may require some trial and error. Also, define the model function.

```
> ωo :='ωo':
  ωoStep := 5e2 :
```

$$y := \frac{Astar}{\left(1 + \frac{\omega^2}{gstar^2}\left(1 - \frac{\omega o^2}{\omega^2}\right)^2\right)^{\frac{1}{2}}};$$

$$y := \frac{0.729649817392542}{\sqrt{1 + 2.33105403776590\ 10^{-10}\ \omega^2 \left(1 - \frac{\omega o^2}{\omega^2}\right)^2}} \quad \textbf{(8)}$$

Now use nested loops to recaluated ChiSq at many values of ωo that bracket ωostar.

```
> ChiSqList := NULL :
  ωoList := NULL :
  for ωo from ωostar − 250·ωoStep to ωostar + 250·ωoStep by ωoStep do;
  ωoList := ωoList, ωo;

  ChiSq := 0 :
  for i from 1 to nops(Vratio) do:
  ω := ωdata[i] :
```

$$ChiSq := ChiSq + \left(\frac{(Vratio[i] - y)}{\sigma Vratio[i]}\right)^2 :$$

```
  end do:
  ChiSqList := ChiSqList, ChiSq;
  ω :='ω':

  end do:
  ωoList := [ωoList] :
  ChiSqList := [ChiSqList] :
```

Now plot the calculated ChiSq values as a function of the parameter ωo. Notice that, over a large range of ωo, ChiSq is not strictly quadratic. However, if we zoom in close to the minimum the dependence is very close to quadratic.

```
> ScatterPlot(ωoList, ChiSqList);
```

$ScatterPlot(\omega oList, ChiSqList, view = [\omega ostar - 15 \cdot \omega oStep .. \omega ostar + 15 \cdot \omega oStep, 50 .. 500]);$





Notice that, in this case, we've actually found a set of parameters that gives a *lower* ChiSqZero than Maple found.

The last step is to characterize the shape of the quadratic dependence:

ChiSq=ChiSqZero+B*(ωo-ωostar)^2

There are three unknowns (ChiSqZero, B, and ωostar). These could be determined simply by fitting the data above to a quadratic function. Alternatively, if we pick three values of ωo near the minimum, we can use the corresponding ChiSq values to determine the unknowns.

> $ChiSq1 := ChiSqList[244];$
$ChiSq2 := ChiSqList[247];$
$ChiSq3 := ChiSqList[250];$
$\Delta\omega o := \omega oList[250] - \omega oList[247];$

$$ChiSq1 := 101.128853653468$$
$$ChiSq2 := 56.8734845219439$$
$$ChiSq3 := 61.1699194058887$$
$$\Delta\omega o := 1500. \quad \textbf{(9)}$$

The value of ωo that minimizes ChiSq (ωoMin) is given below as is the uncertainty in ωoMin which is determined by how quickly ChiSq increase as we move away from ωoMin. That is, σωo is determined by the B in the quadratic dependence of ChiSq on ωo. The larger the value of B, the lower the σωo.

> $\omega oMin := \omega oList[250] - \Delta\omega o \cdot \left(\frac{ChiSq3 - ChiSq2}{ChiSq1 - 2 \cdot ChiSq2 + ChiSq3} + \frac{1}{2} \right);$

$\sigma\omega o := evalf\left(\frac{\mathrm{sqrt}(2) \cdot \Delta\omega o}{\mathrm{sqrt}(ChiSq1 - 2 \cdot ChiSq2 + ChiSq3)} \right);$

$ChiSqZeroFit := ChiSq3 - \frac{1}{\sigma\omega o^2} \cdot (\omega oList[250] - \omega oMin)^2;$

$$\omega oMin := 2.13753098374103\ 10^5$$

$$\sigma\omega o := 304.441302317860$$

$$ChiSqZeroFit := 52.7626269021716 \quad \textbf{(10)}$$

>

Finally, one would report the following for the experimental determination of ωo:

ωo = (2.138 +/- 0.003)*10^5 rad/s

In general, extracting uncertainties in the fit parameters from a nonlinear fit is nontrivial. Above we have given one method for obtaining an **estimate** of the uncertainties in the fit parameters. For more details about uncertainties for nonlinear fits, see **Data Reduction and Error Analysis** by Bevington and Robinson and/or **Numerical Recipes in C** which you can obtain and read for free by visiting http://www.nrbook.com/a/bookcpdf.php.